I - Définition des polynômes de Chebyshev

1. Premières propriétés

1. a. La fonction T du package orthopoly

Le package `orthopoly` définit les principales familles de polynômes orthogonaux :

```
> restart: with(orthopoly);
```

$$[G, H, L, P, T, U]$$

Voici les polynômes de Chebyshev de première espèce, d'indice 0 à 9. On devine quelques propriétés élémentaires des polynômes T_k : degré, parité, coefficient du terme dominant...

```
> matrix(5,2,(i,j)->T[2*i+j-3]=T(2*i+j-3,x));
```

$$\begin{bmatrix} T_0 = 1 & T_1 = x \\ T_2 = 2\,x^2 - 1 & T_3 = 4\,x^3 - 3\,x \\ T_4 = 8\,x^4 - 8\,x^2 + 1 & T_5 = 16\,x^5 - 20\,x^3 + 5\,x \\ T_6 = 32\,x^6 - 48\,x^4 + 18\,x^2 - 1 & T_7 = 64\,x^7 - 112\,x^5 + 56\,x^3 - 7\,x \\ T_8 = 128\,x^8 - 256\,x^6 + 160\,x^4 - 32\,x^2 + 1 & T_9 = 256\,x^9 - 576\,x^7 + 432\,x^5 - 120\,x^3 + 9\,x \end{bmatrix}$$

La syntaxe d'appel de la fonction `T` est `T(n,x)`. Quand l'argument x est lui même une expression, le résultat final n'est pas développé, comme on le voit sur l'exemple suivant :

```
> T(5,x),T(5,a-1/a);
```

$$16\,x^5 - 20\,x^3 + 5\,x string;,\,string; 16\left(a - \frac{1}{a}\right)^5 - 20\left(a - \frac{1}{a}\right)^3 + 5\,a - \frac{5}{a}$$

1. b. Quelques relations entre les polynômes T[n]

L'utilisation de la fonction `T` du package `orthopoly` permet de vérifier certaines relations, pour des valeurs particulières de n. La fonction `eq` est un simple utilitaire qui nous permettra d'afficher les relations souhaitées, de manière plus convaincante. Cette fonction sera beaucoup utilisée dans la suite de ce chapitre.

```
> eq:=proc(obj,opr)
   print(obj=value(obj)); if nargs>1 then obj=opr(value(obj)); fi;
  end :
```

Ici, on vérifie l'égalité $T_n = 2xT_{n-1} - T_{n-2}$ pour $n = 7$:

```
> n:=7: eq('T(n,x)'); eq('2*x*T(n-1,x)-T(n-2,x)',expand);
```

$$\mathrm{T}(n,\,x) = 64\,x^7 - 112\,x^5 + 56\,x^3 - 7\,x$$

$$2\,x\,\mathrm{T}(n-1,\,x) - \mathrm{T}(n-2,\,x) = 2\,x\,(32\,x^6 - 48\,x^4 + 18\,x^2 - 1) - 16\,x^5 + 20\,x^3 - 5\,x$$
$$2\,x\,\mathrm{T}(n-1,\,x) - \mathrm{T}(n-2,\,x) = 64\,x^7 - 112\,x^5 + 56\,x^3 - 7\,x$$

On vérifie maintenant l'égalité $T_{m+n}+T_{m-n}=2T_mT_n$, avec par exemple $m=7$ et $n=3$:

```
> m:=7; n:=3;
  eq('2*T(m,x)*T(n,x)',expand); eq('T(m+n,x)');
  eq('T(m-n,x)'); eq('T(m+n,x)+T(m-n,x)');
```

$$m := 7$$

$$n := 3$$

$$2\,\mathrm{T}(m,\,x)\,\mathrm{T}(n,\,x) = 2\,(64\,x^7 - 112\,x^5 + 56\,x^3 - 7\,x)\,(4\,x^3 - 3\,x)$$

$$2\,\mathrm{T}(m,\,x)\,\mathrm{T}(n,\,x) = 512\,x^{10} - 1280\,x^8 + 1120\,x^6 - 392\,x^4 + 42\,x^2$$

$$\mathrm{T}(m+n,\,x) = 512\,x^{10} - 1280\,x^8 + 1120\,x^6 - 400\,x^4 + 50\,x^2 - 1$$

$$\mathrm{T}(m-n,\,x) = 8\,x^4 - 8\,x^2 + 1$$

$$\mathrm{T}(m+n,\,x) + \mathrm{T}(m-n,\,x) = 512\,x^{10} - 1280\,x^8 + 1120\,x^6 - 392\,x^4 + 42\,x^2$$

De même on vérifie que $T_m(T_n)=T_n(T_m)=T_{mn}$, avec ici $m=3$ et $n=4$:

```
> m:=3; n:=4;
  eq('T(m*n,x)'); eq('T(m,T(n,x))',expand); eq('T(n,T(m,x))',expand);
```

$$m := 3$$

$$n := 4$$

$$\mathrm{T}(m\,n,\,x) = 2048\,x^{12} - 6144\,x^{10} + 6912\,x^8 - 3584\,x^6 + 840\,x^4 - 72\,x^2 + 1$$

$$\mathrm{T}(m,\,\mathrm{T}(n,\,x)) = 4\,(8\,x^4 - 8\,x^2 + 1)^3 - 24\,x^4 + 24\,x^2 - 3$$

$$\mathrm{T}(m,\,\mathrm{T}(n,\,x)) = 2048\,x^{12} - 6144\,x^{10} + 6912\,x^8 - 3584\,x^6 + 840\,x^4 - 72\,x^2 + 1$$

$$\mathrm{T}(n,\,\mathrm{T}(m,\,x)) = 8\,(4\,x^3 - 3\,x)^4 - 8\,(4\,x^3 - 3\,x)^2 + 1$$

$$\mathrm{T}(n,\,\mathrm{T}(m,\,x)) = 2048\,x^{12} - 6144\,x^{10} + 6912\,x^8 - 3584\,x^6 + 840\,x^4 - 72\,x^2 + 1$$

Enfin on vérifie $T_{2n}(x)=2T_n^2(x)-1=T_n(2x^2-1)$, avec $n=5$:

```
> n:=5;
  eq('T(n,x)'); eq('T(2*n,x)');
  eq('2*T(n,x)^2-1',expand);
  eq('T(n,2*x^2-1)',expand);
```

$$n := 5$$

$$\mathrm{T}(n,\,x) = 16\,x^5 - 20\,x^3 + 5\,x$$

$$\mathrm{T}(2\,n,\,x) = 512\,x^{10} - 1280\,x^8 + 1120\,x^6 - 400\,x^4 + 50\,x^2 - 1$$

$$2\,\mathrm{T}(n,\,x)^2 - 1 = 2\,(16\,x^5 - 20\,x^3 + 5\,x)^2 - 1$$

$$2\,\mathrm{T}(n,\,x)^2 - 1 = 512\,x^{10} - 1280\,x^8 + 1120\,x^6 - 400\,x^4 + 50\,x^2 - 1$$

$$\mathrm{T}(n,\,2\,x^2 - 1) = 16\,(2\,x^2 - 1)^5 - 20\,(2\,x^2 - 1)^3 + 10\,x^2 - 5$$

$$\mathrm{T}(n,\,2\,x^2 - 1) = 512\,x^{10} - 1280\,x^8 + 1120\,x^6 - 400\,x^4 + 50\,x^2 - 1$$

1. c. Divisions euclidiennes entre polynômes de Chebyshev

L'instruction suivante montre, pour $n = 9$, les divisions euclidiennes du polynôme T_n par le polynôme T_m, avec $m = n, m = n - 1, \ldots, m = 0$.

On remarque qu'au début le quotient $Q_{n,m}$ est égal à $2T_{n-m}$ et que le reste est égal à $-T_{2m-n}$.

On voit que ce n'est plus aussi simple quand m devient égal ou inférieur à $\dfrac{n}{3}$...

En effet le quotient de T_9 par T_3 n'est pas $2T_6$ mais $Q_{9,3} = 2T_6 - 1$, et le reste est nul.

```
> n:=9;
  convert([seq([Q[n,n-k]=quo(T(n,x),T(n-k,x),x,'r'),
                R[n,n-k]=r],k=1..9)],matrix);
```

$$n := 9$$

$$\begin{bmatrix} Q_{9,8} = 2x & R_{9,8} = -64x^7 + 112x^5 - 56x^3 + 7x \\ Q_{9,7} = 4x^2 - 2 & R_{9,7} = -16x^5 + 20x^3 - 5x \\ Q_{9,6} = 8x^3 - 6x & R_{9,6} = -4x^3 + 3x \\ Q_{9,5} = 16x^4 - 16x^2 + 2 & R_{9,5} = -x \\ Q_{9,4} = 32x^5 - 40x^3 + 10x & R_{9,4} = -x \\ Q_{9,3} = 64x^6 - 96x^4 + 36x^2 - 3 & R_{9,3} = 0 \\ Q_{9,2} = 128x^7 - 224x^5 + 104x^3 - 8x & R_{9,2} = x \\ Q_{9,1} = 256x^8 - 576x^6 + 432x^4 - 120x^2 + 9 & R_{9,1} = 0 \\ Q_{9,0} = 256x^9 - 576x^7 + 432x^5 - 120x^3 + 9x & R_{9,0} = 0 \end{bmatrix}$$

On confirme ici avec $m = 4$ que T_{3m} est divisible par T_m, et que le quotient est $4T_m^2 - 3$.

```
> m:=4; eq('T(3*m,x)/T(m,x)',simplify); eq('4*T(m,x)^2-3',expand);
```

$$m := 4$$

$$\frac{\mathrm{T}(3m, x)}{\mathrm{T}(m, x)} = \frac{2048x^{12} - 6144x^{10} + 6912x^8 - 3584x^6 + 840x^4 - 72x^2 + 1}{8x^4 - 8x^2 + 1}$$

$$\frac{\mathrm{T}(3m, x)}{\mathrm{T}(m, x)} = 256x^8 - 512x^6 + 320x^4 - 64x^2 + 1$$

$$4\,\mathrm{T}(m, x)^2 - 3 = 4(8x^4 - 8x^2 + 1)^2 - 3$$

$$4\,\mathrm{T}(m, x)^2 - 3 = 256x^8 - 512x^6 + 320x^4 - 64x^2 + 1$$

Les procédures `Q` et `R` forment le quotient $Q_{n,m}$ et le reste $R_{n,m}$ dans la division de T_n par T_m.
Tous les cas particuliers sont traités (n multiple de m par un entier impair, par exemple).
Les entiers n et m doivent être tous les deux strictement positifs.

```
> Q:=proc(n::posint,m::posint)
   local k,p; p:=(n-m)/(2*m); 2*sum((-1)^k*T(n-(2*k+1)*m,x),k=0..p);
   if type(p,integer) then %-(-1)^p else % fi;
  end :
  R:=proc(n::posint,m::posint)
   local q,p; q:=(n-m)/(2*m); p:=ceil(q);
   if type(q,integer) then 0 else (-1)^p*T(abs(n-2*p*m),x) fi
  end :
```

On vérifie que `Q` donne le bon résultat, avec par exemple $n = 15$, et $m \in \{12, 5, 3, 2\}$.
Pour cela on compare les quotients calculés par `Q` et par la fonction intégrée `quo` :

```
> n:=15 :
  for m in [12,5,3,2] do eq('Q'(n,m));
    eq('quo'('T'(n,x),'T'(m,x),x));
  od;
```

$$\mathrm{Q}(15,\ 12) = 8\,x^3 - 6\,x$$

$$\mathrm{quo}(\mathrm{T}(15,\ x),\ \mathrm{T}(12,\ x),\ x) = 8\,x^3 - 6\,x$$

$$\mathrm{Q}(15,\ 5) = 1024\,x^{10} - 2560\,x^8 + 2240\,x^6 - 800\,x^4 + 100\,x^2 - 3$$

$$\mathrm{quo}(\mathrm{T}(15,\ x),\ \mathrm{T}(5,\ x),\ x) = 1024\,x^{10} - 2560\,x^8 + 2240\,x^6 - 800\,x^4 + 100\,x^2 - 3$$

$$\mathrm{Q}(15,\ 3) = 4096\,x^{12} - 12288\,x^{10} + 13824\,x^8 - 7232\,x^6 + 1776\,x^4 - 180\,x^2 + 5$$

$$\mathrm{quo}(\mathrm{T}(15,\ x),\ \mathrm{T}(3,\ x),\ x) = 4096\,x^{12} - 12288\,x^{10} + 13824\,x^8 - 7232\,x^6 + 1776\,x^4 - 180\,x^2 + 5$$

$$\mathrm{Q}(15,\ 2) = 8192\,x^{13} - 26624\,x^{11} + 32768\,x^9 - 18816\,x^7 + 4992\,x^5 - 528\,x^3 + 16\,x$$

$$\mathrm{quo}(\mathrm{T}(15, x), \mathrm{T}(2, x), x) = 8192\,x^{13} - 26624\,x^{11} + 32768\,x^9 - 18816\,x^7 + 4992\,x^5 - 528\,x^3 + 16\,x$$

On effectue la même vérification avec la procédure `R` :

```
> n:=15 :
  for m in [12,5,3,2] do
    eq('R'(n,m)); eq('rem'('T'(n,x),'T'(m,x),x));
  od;
```

$$\mathrm{R}(15,\ 12) = -256\,x^9 + 576\,x^7 - 432\,x^5 + 120\,x^3 - 9\,x$$

$$\mathrm{rem}(\mathrm{T}(15,\ x),\ \mathrm{T}(12,\ x),\ x) = -256\,x^9 + 576\,x^7 - 432\,x^5 + 120\,x^3 - 9\,x$$

$$\mathrm{R}(15,\ 5) = 0$$

$$\mathrm{rem}(\mathrm{T}(15,\ x),\ \mathrm{T}(5,\ x),\ x) = 0$$

$$\mathrm{R}(15,\ 3) = 0$$

$$\mathrm{rem}(\mathrm{T}(15,\ x),\ \mathrm{T}(3,\ x),\ x) = 0$$

$$\mathrm{R}(15,\ 2) = x$$

$$\mathrm{rem}(\mathrm{T}(15,\ x),\ \mathrm{T}(2,\ x),\ x) = x$$

On peut aussi, si on n'est encore pas convaincu, vérifier le fonctionnement de `Q` et `R` pour les couples (n,m) tels que $0<m<n$, avec une valeur fixée de n, par exemple $n=50$.

Les deux valeurs logiques obtenues ci-dessous confirment que tous les résultats sont corrects :

```
> n:=50: true: for m from n-1 to 1 by -1 do
    % and evalb(Q(n,m)=quo(T(n,x),T(m,x),x))
  od: %;
  true: for m from n+1 to 1 by -1 do
    % and evalb(R(n,m)=rem(T(n,x),T(m,x),x))
  od: %;
```

$$true$$

$$true$$

2. Définitions possibles de T[n]

2. a. Trois méthodes basées sur le développement de cos(n arccos(x))

Voici une première manière de retrouver les polynômes de Chebyshev T$[n]$. On utilise ici l'expression résultant de l'application de la formule de Moivre, et on demande à Maple de finaliser le développement.

```
> x:='x': Ta:=(n::nonnegint,x)->
      expand(sum(binomial(n,2*'k')*(x^2-1)^'k'*x^(n-2*'k'),'k'=0..n/2));
```

$$Ta := (n{::}nonnegint,\, x) \to \operatorname{expand}\left(\sum_{\text{'}k\text{'}=0}^{1/2\,n} \operatorname{binomial}(n,\, 2\,\text{'}k\text{'})\,(x^2-1)^{\text{'}k\text{'}}\,x^{(n-2\,\text{'}k\text{'})}\right)$$

On vérifie que le résultat est correct avec le polynôme d'indice 5 :

```
> Ta(5,x);
```

$$16\,x^5 - 20\,x^3 + 5\,x$$

On pourrait également, toujours en utilisant la formule de Moivre, écrire que T$[n]$ est la partie réelle de $\exp(i \arccos(x))^n$, c'est-à-dire celle de $(x+i\sqrt{1-x^2})^n$.

Voici une proposition de solution.

On compte ici sur la fonction `expand` pour développer la puissance n-ième :

```
> Tb:=(n::nonnegint,x)->Re(expand((x+I*sqrt(1-x^2))^n));
```

$$Tb := (n{::}nonnegint,\, x) \to \Re(\operatorname{expand}((x + I\,\sqrt{1-x^2})^n))$$

Mais on voit que la fonction `Re` n'agit pas sur l'expression développée.
Maple ne sait en effet pas ce que contient x (son contenu pourrait être un nombre complexe) :

```
> Tb(4,x);
```

$$1 + 4\,\Re(2\,x^4 + 2\,I\,\sqrt{1 - x^2}\,x^3 - 2\,x^2 - I\,\sqrt{1 - x^2}\,x)$$

Même en supposant que x est réel, on se heurte à une difficulté provenant de la racine carrée de $1 - x^2$:

```
> assume(x,real): Tb(4,x);
```

$$8\,x^4 - 8\,x^2 + 1 + 4\,\Re(2\,I\,\sqrt{1 - x^2}\,x^3 - I\,\sqrt{1 - x^2}\,x)$$

Il faut donc supposer que x est compris entre -1 et 1 :

```
> assume(x,RealRange(-1,1)): Tb(4,x);
```

$$8\,x^4 - 8\,x^2 + 1$$

Voici maintenant une troisième façon de retrouver les polynômes de Chebyshev T_n.
On utilise la définition $T_n(x) = \cos(n \arccos(x))$ et on demande à Maple de développer :

```
> x:='x': Tc:=(n::nonnegint,x)->expand(cos(n*arccos(x)));
```

$$Tc := (n::nonnegint,\, x) \to \operatorname{expand}(\cos(n \arccos(x)))$$

On retrouve bien le polynôme T_5, par exemple :

```
> Tc(5,x);
```

$$16\,x^5 - 20\,x^3 + 5\,x$$

Il se passe cependant une chose étonnante : on peut ainsi calculer tous les T_n jusqu'à $n = 99$ (et le résultat est instantané), mais ce n'est plus possible à partir de $n = 100$, car alors Maple refuse de développer !

```
> eq('expand(cos(10*theta))'); eq('expand(cos(100*theta))');
```

$$\operatorname{expand}(\cos(10\,\theta)) = 512\cos(\theta)\hat{}10 - 1280\cos(\theta)\hat{}8 + 1120\cos(\theta)\hat{}6 - 400\cos(\theta)\hat{}4 + 50\cos(\theta)\hat{}2 - 1$$

$$\operatorname{expand}(\cos(100\,\theta)) = \cos(100\,\theta)$$